Computational Companion to “Quasi-Symmetric Nets: A Constructive Approach to the Equimodular Elliptic Type of Kokotsakis Polyhedra” — CRITERION HELPER

Author: Abdukhomid Nurmatov
Tested on: Mathematica 14.0

Section 1

Results of Section 1 is used in Lemma 1

In[1]:=
```
(*Clear all previous definitions*)
ClearAll[α, β, γ, δ, σ, ᾱ, β̄, γ̄, δ̄, a, b, c, d, M]

(*Define the angles and variables*)
σ = (α + β + γ + δ) / 2;
ᾱ = σ - α;
β̄ = σ - β;
γ̄ = σ - γ;
δ̄ = σ - δ;

a = Sin[α] / Sin[ᾱ];
b = Sin[β] / Sin[β̄];
c = Sin[γ] / Sin[γ̄];
d = Sin[δ] / Sin[δ̄];
M = a * b * c * d;
```

```
(*Define the 7 identities*)
expr1 = 1 - a * b == Sin[σ] * Sin[ᾱ - β] / (Sin[ᾱ] * Sin[β̄]);
expr2 = 1 - b * c == Sin[σ] * Sin[γ̄ - β] / (Sin[γ̄] * Sin[β̄]);
expr3 = 1 - b * d == Sin[σ] * Sin[δ̄ - β] / (Sin[δ̄] * Sin[β̄]);
expr4 = c * d - 1 == Sin[σ] * Sin[ᾱ - β] / (Sin[γ̄] * Sin[δ̄]);
expr5 = a * d - 1 == Sin[σ] * Sin[γ̄ - β] / (Sin[ᾱ] * Sin[δ̄]);
expr6 = a * c - 1 == Sin[σ] * Sin[δ̄ - β] / (Sin[ᾱ] * Sin[γ̄]);
expr7 = 1 - M ==
   Sin[σ] * Sin[ᾱ - β] * Sin[γ̄ - β] * Sin[δ̄ - β] / (Sin[ᾱ] * Sin[β̄] * Sin[γ̄] * Sin[δ̄]);

(*Store the expressions and labels*)
lhs = {"1 - ab", "1 - bc", "1 - bd", "cd - 1", "ad - 1", "ac - 1", "1 - M"};
rhs =
  {"sin(σ)sin(ᾱ - β) / (sin(ᾱ)sin(β̄))", "sin(σ)sin(γ̄ - β) / (sin(β̄)sin(γ̄))",
   "sin(σ)sin(δ̄ - β) / (sin(β̄)sin(δ̄))", "sin(σ)sin(ᾱ - β) / (sin(γ̄)sin(δ̄)",
   "sin(σ)sin(γ̄ - β) / (sin(ᾱ)sin(δ̄))", "sin(σ)sin(δ̄ - β) / (sin(ᾱ)sin(γ̄))",
   "sin(σ)sin(ᾱ - β)sin(γ̄ - β)sin(δ̄ - β) / (sin(ᾱ)sin(β̄)sin(γ̄)sin(δ̄))"};

expressions = {expr1, expr2, expr3, expr4, expr5, expr6, expr7};

(*Function to test equality*)
isTrueQ[expr_] := TrueQ[FullSimplify[expr]]

(*Build the result table*)
TableForm[Table[{lhs⟦i⟧, rhs⟦i⟧, If[isTrueQ[expressions⟦i⟧], "✓", "✗"]},
  {i, Length[expressions]}],
 TableHeadings → {None, {"LHS", "RHS", "LHS = RHS?"}}]
```

Out[23]//TableForm=

LHS	RHS	LI
1 - ab	sin(σ)sin(ᾱ - β) / (sin(ᾱ)sin(β̄))	✓
1 - bc	sin(σ)sin(γ̄ - β) / (sin(β̄)sin(γ̄))	✓
1 - bd	sin(σ)sin(δ̄ - β) / (sin(β̄)sin(δ̄))	✓
cd - 1	sin(σ)sin(ᾱ - β) / (sin(γ̄)sin(δ̄)	✓
ad - 1	sin(σ)sin(γ̄ - β) / (sin(ᾱ)sin(δ̄))	✓
ac - 1	sin(σ)sin(δ̄ - β) / (sin(ᾱ)sin(γ̄))	✓
1 - M	sin(σ)sin(ᾱ - β)sin(γ̄ - β)sin(δ̄ - β) / (sin(ᾱ)sin(β̄)sin(γ̄)sin(δ̄))	✓

Section 2

Results of Section 2 is used in Proof of Lemma 6

In[65]:=
```
(*Clear all previous definitions*)
ClearAll[α, β, γ, δ, σ, ᾱ, β̄, γ̄, δ̄, u, x, y, z, d1, d2, d3, d4, d5,
```

```
 rhsCosAlpha, rhsCosBeta, rhsCosGamma, rhsCosDelta, rhsCosSigma]

(*Define assumptions for the angles*)
assumptions = {0 < α < Pi, 0 < β < Pi, 0 < γ < Pi,
   0 < δ < Pi, 0 < ᾱ < Pi, 0 < β̄ < Pi, 0 < γ̄ < Pi, 0 < δ̄ < Pi};

(*Define the angles and intermediate variables*)
σ = (α + β + γ + δ) / 2;
ᾱ = σ - α;
β̄ = σ - β;
γ̄ = σ - γ;
δ̄ = σ - δ;
ε = Abs[Sin[σ]] / Sin[σ];

(*Define section 1 results*)
OneMinusM =
  Sin[σ] * Sin[ᾱ - β] * Sin[γ̄ - β] * Sin[δ̄ - β] / (Sin[ᾱ] * Sin[β̄] * Sin[γ̄] * Sin[δ̄]);
rMinusOne = Sin[σ] * Sin[γ̄ - β] / (Sin[ᾱ] * Sin[δ̄]);
sMinusOne = Sin[σ] * Sin[ᾱ - β] / (Sin[γ̄] * Sin[δ̄]);
fMinusOne = Sin[σ] * Sin[δ̄ - β] / (Sin[ᾱ] * Sin[γ̄]);

u = OneMinusM;
x = 1 / rMinusOne;
y = 1 / sMinusOne;
z = 1 / fMinusOne;

(*Define expressions in denominators*)
d1 = x * z * u * (1 + y) * (1 + u * y);
d2 = y * z * u * (1 + x) * (1 + u * x);
d3 = x * y * u * (1 + z) * (1 + u * z);
d4 = u * (1 + x) * (1 + y) * (1 + z) * (1 + u * x) * (1 + u * y) * (1 + u * z);
d5 = x * y * z * u ^ 2 * (1 + x) * (1 + y) * (1 + z);

(*Define denominators*)
denAlpha = 2 * FullSimplify[Sqrt[d1], assumptions];
denBeta = 2 * FullSimplify[Sqrt[d4], assumptions];
denGamma = 2 * FullSimplify[Sqrt[d2], assumptions];
denDelta = 2 * FullSimplify[Sqrt[d3], assumptions];
denSigma = 2 * FullSimplify[Sqrt[d5], assumptions];

(*Define RHS expressions*)
rhsCosAlpha = FullSimplify[ε * (1 - y * z * u + x * z * u - x * y * u) / denAlpha] /.
   Cos[β] - Cos[α + γ + δ] :> 2 Sin[(α + β + γ + δ) / 2] Sin[(α - β + γ + δ) / 2];

rhsCosBeta =
  FullSimplify[ε * (u * (1 + x) * (1 + y) * (1 + z) + (1 + u * x) * (1 + u * y) * (1 + u * z) -
```

```
        u * x * y * z * (u - 1) ^ 2) / denBeta] /.
    Cos[α] - Cos[β + γ + δ] :> 2 Sin[(α + β + γ + δ) / 2] Sin[(-α + β + γ + δ) / 2];

rhsCosGamma = FullSimplify[ε * (1 + y * z * u - x * z * u - x * y * u) / denGamma] /.
    Cos[β] - Cos[α + γ + δ] :> 2 Sin[(α + β + γ + δ) / 2] Sin[(α - β + γ + δ) / 2];

rhsCosDelta = FullSimplify[ε * (1 - y * z * u - x * z * u + x * y * u) / denDelta] /.
    Cos[β] - Cos[α + γ + δ] :> 2 Sin[(α + β + γ + δ) / 2] Sin[(α - β + γ + δ) / 2];

rhsCosSigma = (1 - u * (x * y + x * z + y * z + 2 * x * y * z)) / denSigma;

(*Define equations to check*)
exprAlpha = Cos[α] == FullSimplify[rhsCosAlpha, assumptions];
exprBeta = Cos[β] == FullSimplify[rhsCosBeta, assumptions];
exprGamma = Cos[γ] == FullSimplify[rhsCosGamma, assumptions];
exprDelta = Cos[δ] == FullSimplify[rhsCosDelta, assumptions];
exprSigma = Cos[σ] == FullSimplify[rhsCosSigma, assumptions];

(*Build labeled table*)
lhs = {"cos(α)", "cos(β)", "cos(γ)", "cos(δ)", "cos(σ)"};
rhs = {"ε(1 - yzu + xzu - xyu) / (2√(xzu(1 + y)(1 + uy)))",
    "ε(u(1 + x)(1 + y)(1 + z) + (1 + ux)(1 + uy)(1 + uz) - uxyz(u -
      1)^2) / (2√(u(1 + x)(1 + y)(1 + z)(1 + ux)(1 + uy)(1 + uz)))",
    "ε(1 + yzu - xzu - xyu) / (2√(yzu(1 + x)(1 + ux)))",
    "ε(1 - yzu - xzu + xyu) / (2√(xyu(1 + z)(1 + uz)))",
    "(1 - u(xy + xz + yz + 2xyz)) / (2√(xyzu^2(1 + x)(1 + y)(1 + z)))"};
expressions = {exprAlpha, exprBeta, exprGamma, exprDelta, exprSigma};

isTrueQ[expr_] := TrueQ[FullSimplify[expr, assumptions]];

TableForm[Table[{lhs[[i]], rhs[[i]], If[isTrueQ[expressions[[i]]], "✓", "✗"]},
   {i, Length[expressions]}],
  TableHeadings → {None, {"LHS", "RHS", "LHS = RHS?"}}]
```

Out[105]//TableForm=

LHS	RHS
cos(α)	ε(1 - yzu + xzu - xyu) / (2√(xzu(1 + y)(1 + uy)))
cos(β)	ε(u(1 + x)(1 + y)(1 + z) + (1 + ux)(1 + uy)(1 + uz) - uxyz(u - 1)^2) /
cos(γ)	ε(1 + yzu - xzu - xyu) / (2√(yzu(1 + x)(1 + ux)))
cos(δ)	ε(1 - yzu - xzu + xyu) / (2√(xyu(1 + z)(1 + uz)))
cos(σ)	(1 - u(xy + xz + yz + 2xyz)) / (2√(xyzu^2(1 + x)(1 + y)(1 + z)))

Section 3

Results of Section 3 is used in Proof of Lemma 7

In[106]:=

```
(*Clear previous definitions*)
ClearAll[α, β, γ, δ, σ, ᾱ, β̄, γ̄, δ̄, ε, θ, A, ξ, η, lhs, rhs, expr, thetaIndex];

(*---Definitions---*)
σ = (α + β + γ + δ) / 2;
ᾱ = σ - α;
β̄ = σ - β;
γ̄ = σ - γ;
δ̄ = σ - δ;
ε = Abs[Sin[σ]] / Sin[σ];

(*Cyclic index function*)
thetaIndex[k_] := Mod[k - 1, 4] + 1;

(*Definition of A[i,j]*)
A[i_, j_] /; 1 ≤ i ≤ 4 && 1 ≤ j ≤ 4 :=
  4 Cos[θ[thetaIndex[i]] / 2 + (Pi / 4) i j (j - 1) + (Pi / 2) j]^2 *
   Cos[θ[thetaIndex[i - 1]] / 2 + (Pi / 4) (i - 1) j (j - 1) + (Pi / 2) j]^2;
(*Table[A[i,j],{i,1,4},{j,1,4}];*)

(*Define ξ[i] and η[i]*)
ξ[i_?IntegerQ] /; 1 ≤ i ≤ 4 :=
  If[OddQ[i], θ[thetaIndex[i]], θ[thetaIndex[i - 1]]];
η[i_?IntegerQ] /; 1 ≤ i ≤ 4 := If[OddQ[i], θ[thetaIndex[i - 1]], θ[thetaIndex[i]]];
(*Table[{i,ξ[i],η[i]},{i,1,4}];*)

(*Define left-hand side and right-hand side expressions*)
lhs[i_Integer] :=
  ε (A[i, 1] Sin[β̄] Sin[σ] + A[i, 2] Sin[γ̄] Sin[γ̄ - β] + A[i, 3] Sin[ᾱ] Sin[ᾱ - β] +
      A[i, 4] Sin[δ̄] Sin[δ̄ - β]) / (2 Sin[α] Sin[γ]);

rhs[i_Integer] :=
  ε (Cos[β] - Cos[γ] (Cos[α] Cos[δ] + Cos[ξ[i]] Sin[α] Sin[δ]) - Cos[η[i]]
       Sin[γ] (Cos[α] Sin[δ] - Cos[ξ[i]] Sin[α] Cos[δ])) / (Sin[α] Sin[γ]);

expr[i_Integer] := FullSimplify[Equal[lhs[i], rhs[i]]];

(*---Build labeled result table---*)
lhsList =
  {"ε₁ (A₁₁sin(β̄₁)sin(σ₁) + A₁₂sin(γ̄₁)sin(γ̄₁ - β₁) + A₁₃sin(ᾱ₁)sin(ᾱ₁ -
```

```
    β₁) + A₁₄sin(δ̄₁)sin(δ̄₁ - β₁)) / (2sin(α₁)sin(γ₁))",
  "ε₂(A₂₁sin(β̄₂)sin(σ₂) + A₂₂sin(γ̄₂)sin(γ̄₂ - β₂) + A₂₃sin(ᾱ₂)sin(ᾱ₂
    - β₂) + A₂₄sin(δ̄₂)sin(δ̄₂ - β₂)) / (2sin(α₂)sin(γ₂))",
  "ε₃(A₃₁sin(β̄₃)sin(σ₃) + A₃₂sin(γ̄₃)sin(γ̄₃ - β₃) + A₃₃sin(ᾱ₃)sin(ᾱ₃
    - β₃) + A₃₄sin(δ̄₃)sin(δ̄₃ - β₃)) / (2sin(α₃)sin(γ₃))",
  "ε₄(A₄₁sin(β̄₄)sin(σ₄) + A₄₂sin(γ̄₄)sin(γ̄₄ - β₄) + A₄₃sin(ᾱ₄)sin(ᾱ₄
    - β₄) + A₄₄sin(δ̄₄)sin(δ̄₄ - β₄)) / (2sin(α₄)sin(γ₄))"};

rhsList = {"ε₁((cos(β₁) - cos(γ₁)(cos(α₁)cos(δ₁) +
    cos(θ₁)sin(α₁)sin(δ₁)) - cos(θ₄)sin(γ₁)(cos(α₁)sin(δ₁)
    - cos(θ₁)sin(α₁)cos(δ₁))) / (sin(α₁)sin(γ₁))",
  "ε₂((cos(β₂) - cos(γ₂)(cos(α₂)cos(δ₂) + cos(θ₁)sin(α₂)sin(δ₂))
    - cos(θ₂)sin(γ₂)(cos(α₂)sin(δ₂) -
    cos(θ₁)sin(α₂)cos(δ₂))) / (sin(α₂)sin(γ₂))",
  "ε₃((cos(β₃) - cos(γ₃)(cos(α₃)cos(δ₃) + cos(θ₃)sin(α₃)sin(δ₃))
    - cos(θ₂)sin(γ₃)(cos(α₃)sin(δ₃) -
    cos(θ₃)sin(α₃)cos(δ₃))) / (sin(α₃)sin(γ₃))",
  "ε₄((cos(β₄) - cos(γ₄)(cos(α₄)cos(δ₄) + cos(θ₃)sin(α₄)sin(δ₄))
    - cos(θ₄)sin(γ₄)(cos(α₄)sin(δ₄) -
    cos(θ₃)sin(α₄)cos(δ₄))) / (sin(α₄)sin(γ₄))"};
results = Table[If[TrueQ[expr[i]], "✓", "✗"], {i, 1, 4}];

TableForm[Table[{i, lhsList⟦i⟧, rhsList⟦i⟧, results⟦i⟧}, {i, 1, 4}],
 TableHeadings → {None, {"i", "LHS", "RHS", "LHS = RHS?"}},
 TableAlignments → Left]
```

Out[123]//TableForm=

i	LHS
1	$\varepsilon_1(A_{11}\sin(\bar{\beta}_1)\sin(\sigma_1) + A_{12}\sin(\bar{\gamma}_1)\sin(\bar{\gamma}_1 - \beta_1) + A_{13}\sin(\bar{\alpha}_1)\sin(\bar{\alpha}_1 - \beta_1) + A_1$
2	$\varepsilon_2(A_{21}\sin(\bar{\beta}_2)\sin(\sigma_2) + A_{22}\sin(\bar{\gamma}_2)\sin(\bar{\gamma}_2 - \beta_2) + A_{23}\sin(\bar{\alpha}_2)\sin(\bar{\alpha}_2 - \beta_2) + A_2$
3	$\varepsilon_3(A_{31}\sin(\bar{\beta}_3)\sin(\sigma_3) + A_{32}\sin(\bar{\gamma}_3)\sin(\bar{\gamma}_3 - \beta_3) + A_{33}\sin(\bar{\alpha}_3)\sin(\bar{\alpha}_3 - \beta_3) + A_3$
4	$\varepsilon_4(A_{41}\sin(\bar{\beta}_4)\sin(\sigma_4) + A_{42}\sin(\bar{\gamma}_4)\sin(\bar{\gamma}_4 - \beta_4) + A_{43}\sin(\bar{\alpha}_4)\sin(\bar{\alpha}_4 - \beta_4) + A_4$